Naval Medical Research Center
Building 503
503 Robert Grant Avenue
Silver Spring, Maryland 20910-7500

NMRC 00-03 July 2000

A MODEL OF DRY DECK SHELTER CO_2 LEVELS

J. F. Himm

Bureau of Medicine and Surgery
Department of the Navy
Washington, DC 20372-5120

NOTICES

TECHNICAL REVIEW AND APPROVAL

NMRC 00-03

The experiments reported herein were conducted according to the principles set forth in the current edition of the "Guide for the Care and Use of Laboratory Animals," Institute of Laboratory Animal Resources, National Research Council, National Academy Press, 1996."

This technical report has been reviewed by the NMRC scientific and public affairs staff and is approved for publication. It is releasable to the National Technical Information Service where it will be available to the general public, including foreign nations.

RICHARD G. HIBBS
CAPT,MC,USN
Commanding Officer
Naval Medical Research Center

REPORT DOCUMENTATION PAGE

Form Approved
OMB No. 0704-0188

1. AGENCY USE ONLY *(Leave Blank)*	2. REPORT DATE	3. REPORT TYPE AND DATES COVERED
	July 2000	Technical 12/98 – 2/99

4. TITLE AND SUBTITLE
A model of dry deck shelter CO_2 levels.

5. FUNDING NUMBERS*
PE – 63713N
PR – M0099
TA – 01C
WU – 1506

6. AUTHOR(S)

Himm, J.F.**

7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES)

NAVAL MEDICAL RESEARCH CENTER
(CODE 00)
503 ROBERT GRANT AVENUE
SILVER SPRING, MARYLAND 20910-7500

8. PERFORMING ORGANIZATION REPORT NUMBER

NMRC 00-03

9. SPONSORING/MONITORING AGENCY NAME(S) AND ADDRESS(ES)

BUREAU OF MEDICINE AND SURGERY
(MED-02)
2300 E STREET, NW
WASHINGTON, DC 20372-5300

10. SPONSORING/MONITORING AGENCY REPORT NUMBER

DN241125

11. SUPPLEMENTARY NOTES

*Also funded by Naval Sea Systems Command (PMS-395)
**Department of Physiology and Biophysics; Georgetown University; Washington, DC

12a. DISTRIBUTION/AVAILABILITY STATEMENT

Approved for public release; distribution is unlimited

12b. DISTRIBUTION CODE

13. ABSTRACT *(Maximum 200 words)*

To maintain a safe environment, the partial pressure of carbon dioxide (CO_2) in a closed space needs to be kept under control. In the dry deck shelter (DDS) the CO_2 pressure is controlled primarily through the use of venting of each of the three modules. Air from the submarine air banks, which contain less than 1250 ppm CO_2, is added to a compartment at the same time that an exhaust valve is open, allowing a constant pressure to be maintained in the compartment. This report will provide details of a model of the CO_2 partial pressure in a DDS. The model assumes that the gas spaces are well mixed, containing a uniform concentration of CO_2 within a given compartment. Each gas space has the following elements as parameters: the number of divers in the compartment, the average rate of CO_2 production by each diver, the ambient pressure in the compartment, the effective gas space volume (accounting for the degree to which the compartment is flooded with water), the temperature of the compartment, the venting cycle time, the flow rate of the venting gas, and the concentration of CO_2 in the supply (ventilation) gas. In addition to a detailed description of the model and its limitations, this report includes details on the application of the model to operational conditions and data for CO_2 levels in a DDS, and the computer code used to make the comparison.

14. SUBJECT TERMS
dry deck shelter, carbon dioxide, venting

15. NUMBER OF PAGES
33

16. PRICE CODE

17. SECURITY CLASSIFICATION OF REPORT	18. SECURITY CLASSIFICATION OF THIS PAGE	19. SECURITY CLASSIFICATION OF ABSTRACT	20. LIMITATION OF ABSTRACT
Unclassified	Unclassified	Unclassified	Unclassified

NSN 7540-01-280-5500

Standard Form 298 (Rev 2-89)
Prescribed by ANSI S10 239-98
298-102

TABLE OF CONTENTS

List of Appendices

List of Tables

ACKNOWLEDGMENTS

This was supported by the Naval Sea Systems Command (PMS-395) and performed under Office of Naval Research work unit 63713N M0099.01C-1506. The opinions and assertions contained herein are the private ones of the author and are not to be construed as official or reflecting the views of the Navy Department or the naval service at large.

BACKGROUND

The dry deck shelter (DDS) is a three-compartment unit, which can be mated to a submarine for operations. These three compartments, the hanger, transfer trunk, and chamber, are described below.

(1) Hanger: With a dry volume ≈ 1400 ft^3, it is the largest of the three compartments. It has a large hatch at the aft end, which allows for the ingress and egress of the divers and their equipment. The hanger can be pressurized independently and flooded with water to various depths.

(2) Transfer Trunk: Connected at the aft end to the hanger, the trunk is mated to the submarine. Transfer of men and equipment between the submarine and the DDS is through a hatch in the floor, and may need to be done when the hanger is flooded and open to the sea. Since the submarine is maintained at a pressure of 1 atmosphere absolute (ATA), the trunk is often cycled between the dry (1 ATA) state of the submarine and the wet (pressurized) state of the hanger. Its pressurization and flood capabilities are the same as the hanger.

(3) Chamber: The chamber is attached to the forward end of the trunk. It is outfitted for DCS treatment and can be independently pressurized to any depth required for treatment.

During operations, varying numbers of divers are present in the compartments, each producing CO_2 and consuming O_2. It is especially important to control the level of CO_2 in the trunk, where there may be 7 or 8 divers breathing in a 50-100 ft^3 air space (when the trunk is flooded). Since the level of CO_2 is not currently measured, this control is accomplished by periodically venting the compartment. Air from the submarine air banks is let into the compartment at the same time as air is being bled out, back into the submarine. This procedure is apparently effective, but has two drawbacks: it requires a large amount of air and is very noisy.

The former makes keeping the air banks charged difficult and the latter hinders communication within the DDS. For both reasons, it is desirable to minimize the amount of venting required.

We will derive the equations that describe the amount of CO_2 within a compartment and then apply those equations to the problem of periodic ventilation of a closed compartment. Finally, we will compare the predictions of the equations with some measurements of CO_2 levels in a DDS from actual Fleet operations.

DERIVATION

We will calculate the partial pressure of CO_2 in a given compartment by determining the number of moles of CO_2 in the compartment, assuming that the ideal gas law holds. The DDS is operated to depths of 165 feet of seawater (fsw), allowing for a maximum gas pressure of 6 ATA. This is within the range of validity of the ideal gas law. If we assume that the respiratory quotient is unity, then each mole of O_2 used by a diver leads to the release of a mole of CO_2. Thus, the only change in the total number of gas molecules in the compartment is due to the flow of air into or out of the compartment.

$$\frac{dn_{tot}}{dt} = F_{in} - F_{out} \equiv F$$

where n_{tot} is the total number of moles of gas in the compartment, and F_{in} and F_{out} are the the number of moles per second flowing into or out of the compartment, respectively. The net flow into the compartment is F.

Carbon dioxide levels in a compartment change due to three different mechanisms: introduction, production, or elimination. Carbon dioxide is introduced by the inflowing air from the submarine air banks, which contain air with an upper limit of 1250 ppm CO_2 (the submarine is usually ventilated using surface air before the air banks are pressurized, so the actual CO_2 concentration is often substantially below this level). Produced by divers that are present in the compartment, CO_2 is eliminated by venting. (We do not consider the elimination of CO_2 by scrubbing. This technique is used on occasion, and only in the chamber portion of the DDS.) Thus, we can write for the number of moles of CO_2

$$\frac{dn_{CO2}}{dt} = C_{in} \cdot F_{in} + N_{diver} \cdot A - \frac{n_{CO2}}{n_{tot}} \cdot F_{out}$$

where n_{CO2} is the number of moles of CO_2 in the compartment, C_{in} is the mole fraction of CO_2 in the inflowing gas, N_{diver} is the number of divers in the compartment, A is the average rate of production of CO_2 by a diver (in moles per second), and the ratio n_{CO2}/n_{tot} is the mole fraction of CO_2 in the gas in the compartment. This final term implies that the gas in the compartment is well-mixed.

There are four cases of interest (Some sub-categories may also be explored below).

a) $F_{out} = 0, \quad F = 0 \Rightarrow F_{in} = 0$
(constant pressure in the compartment with no venting)

b) $F_{out} = 0, \quad F \neq 0 \Rightarrow F_{in} \neq 0$
(pressure in the compartment is increasing)

c) $F_{out} > 0, \quad F = 0 \Rightarrow F_{in} = F_{out}$
(venting the compartment at constant pressure)

d) $F_{out} > 0, \quad F \neq 0 \Rightarrow F_{in} \neq F_{out}$
(venting the compartment while changing the pressure)

In order to simplify the discussion below, we make the following definitions:

$x_1 = n_{tot}$
$x_2 = n_{CO2}$
$B = C_{in} * F_{in} + N_{diver} * A$
$C = F_{out}$

The above equations then become

$$\frac{dx_1 1}{dt} = F$$
$$\frac{dx_2}{dt} = B - \frac{x_2}{x_1} \cdot C$$

Case a) $F_{out} = F_{in} = F = 0 \Rightarrow C = 0$

$$\frac{dx_1}{dt} = 0 \Rightarrow x_1 = x_{10}$$
$$\frac{dx_2}{dt} = B \Rightarrow x_2 = x_{20} + B \cdot t$$

where x_{10} and x_{20} are the initial values of x_1 and x_2, respectively. Thus, the total number of moles of gas is constant and the number of moles of CO_2 increases linearly in time. Since $F_{in} = 0$, the rate of increase of CO_2 depends only upon the number of divers and their rate of CO_2 production ($B = N_{diver} * A$).

Case b) $F_{out} = 0 = C, \quad F = F_{in} \neq 0$

$$\frac{dx_1}{dt} = F_{in} \Rightarrow x_1 = x_{10} + F_{in} \cdot t$$
$$\frac{dx_2}{dt} = B \Rightarrow x_2 = x_{20} + B \cdot t = x_{20} + \left(C_{in} \cdot F_{in} + N_{diver} \cdot A\right) \cdot t$$

Thus, with only inflowing gas both the total number of moles of gas and the number of moles of CO_2 increase linearly as a function of time.

Case c) $F_{out} > 0, \quad F = 0 \Rightarrow F_{in} = F_{out}$

$$\frac{dx_1}{dt} = 0 \Rightarrow x_1 = x_{10}$$

$$\frac{dx_2}{dt} = B - \frac{x_2}{x_{10}} \cdot C \Rightarrow x_2 = \frac{B}{C} \cdot x_{10} \cdot \left(1 - e^{-\frac{C}{x_{10}} \cdot t} \right) + x_{20} \cdot e^{-\frac{C}{x_{10}} \cdot t}$$

This is the case where the compartment is being ventilated. Thus the mole fraction of CO_2 in the compartment exponentially approaches the ratio B/C, where

$$\frac{B}{C} = \frac{C_{in} \cdot F_{in} + N_{diver} \cdot A}{F_{out}}$$

The higher the rates of production and introduction relative to elimination, the higher the asymptotic level of CO_2

Case d) $F_{out} > 0$, $F \neq 0 \Rightarrow F_{in} \neq F_{out}$

$$\frac{dx_1}{dt} = F \Rightarrow x_1 = x_{10} + F \cdot t$$

$$\frac{dx_2}{dt} = B - C \cdot \frac{x_2}{x_1} = B - \frac{C \cdot x_2}{x_{10} + F \cdot t}$$

We can solve this equation for x_2 by changing variables. First, let $t' = x_{10} + F*t$, so that $dt' = F*dt$, and

$$\frac{dx_2}{dt'} = B' - C' \cdot \frac{x_2}{t'}$$

where $B' = B/F$ and $C' = C/F$. This equation is now separable. If we do a change of variable a second time, defining $v = x_2/t'$, then this equation becomes

$$\frac{dt'}{t'} = \frac{dv}{B' - (1 + C') \cdot v}$$

Consider two cases, $F_{in} = 0$ and $F_{in} > 0$.

Subcase d1) $F_{in} = 0 \Rightarrow C' = C/F = F_{out}/(F_{in} - F_{out}) = -1$, so that

$$\frac{dt'}{t'} = \frac{dv}{B'}$$
$$or$$
$$v = B' \cdot \ln(t') + v_0$$

Substituting for v and t' from above yields

$$\frac{x_2}{x_{10}+F \cdot t} = \frac{B}{F} \cdot \ln(x_{10} + F \cdot t) + v_0$$
$$x_2 = \left(\frac{B}{F} \cdot \ln(x_{10} + F \cdot t) + v_0\right) \cdot (x_{10} + F \cdot t)$$

At time $t = 0$, $x_2 = x_{20}$, so that

$$x_2 = \left(\frac{N_{diver} \cdot A}{F_{out}} \cdot \ln\left(\frac{x_{10}}{x_{10} - F_{out} \cdot t}\right) + \frac{x_{20}}{x_{10}}\right)\left(x_{10} - F_{out} \cdot t\right)$$

Subcase d2) $F_{in} > 0$, $F \neq 0$, so that

$$\frac{dt'}{t'} = \frac{dv}{B'-(1+C')\cdot v}$$
$$\ln(t') = \frac{-1}{(1+C')} \cdot \ln(B'-(1+C') \cdot v) + D_0$$

where D_0 is a constant of integration. Substituting in $v = x_2/t'$ and rearranging gives

$$x_2 = \frac{B'}{1+C'} \cdot t' - D_0 \cdot t'^{-C'}$$

At time $t = 0$, $t' = x_{10}$, $x_2 = x_{20}$, and

$$D_0 = \left(\frac{B'}{1+C'} \cdot x_{10} - x_{20}\right) \cdot x_{10}^{\ C'}$$

Then

$$x_2 = \frac{B'}{1+C'} \cdot t' - \left(\frac{B'}{1+C'} \cdot x_{10} - x_{20}\right) \cdot \left(\frac{x_{10}}{t'}\right)^{C'}$$

Upon substitution back to the original variables

$$x_2 = \frac{B}{F_{in}} \cdot (x_{10} - (F_{in} - F_{out}) \cdot t) - (\frac{B}{F_{in}} \cdot x_{10} - x_{20}) \cdot \left(\frac{x_{10}}{x_{10} - (F_{in} - F_{out}) \cdot t} \right)^{\frac{F_{out}}{F_{in} - F_{out}}}$$

where B = N_{diver}*A + C_{in}*F_{in} . Note that this case corresponds to the simultaneous venting of the chamber while the pressure is changing (assuming that the volume of the compartment is fixed).

The partial pressure of CO_2 can now be determined using the ideal gas law

$$P_{CO_2} = \frac{R_g \cdot T}{V} \cdot x_2$$

where R_g is the gas constant, T is the absolute temperature, and V is the gas volume.

VENTING

During operations, venting usually takes place when other factors, N_{diver} , V, and P (total gas pressure), are being held constant. Thus, between vents, when there is no flow into or out of the compartment, Case a) above describes the buildup of CO_2 in the compartment. During venting, the inflow and outflow are balanced (constant P), and Case c) applies. Ventilation occurs periodically, repeating after time t_{cycle} , with ventilation lasting for time t_{vent}. Define

$$E = e^{-\frac{F_{out}}{x_{10}} \cdot t_{vent}}$$
$$E' = 1 - E$$

Then the number of moles of CO_2 at the end of a vent is given by

$$x_2^{off} = \frac{N_{diver} \cdot A + C_{in} \cdot F_{in}}{F_{out}} \cdot x_{10} \cdot E' + x_2^{on} \cdot E$$

where the superscript indicates that the venting gas was turned off or on at that point in time. During the nonventing period, Case a) applies. Therefore, at the end of a non-venting period we can write

$$x_2^{on} = x_2^{off} + N_{diver} \cdot A \cdot (t_{cycle} - t_{vent})$$

We can combine these equations to generate recursive relations describing the CO_2 level at the beginning and end of each vent cycle in terms of the levels of the previous cycle. So,

$$x_{2,i}^{off} = \frac{N_{diver} \cdot A + C_{in} \cdot F_{in}}{F_{out}} \cdot x_{10} \cdot E' + \left(x_{2,i-1}^{off} + N_{diver} \cdot A \cdot (t_{cycle} - t_{vent}) \right) \cdot E$$

$$x_{2,i}^{on} = \frac{N_{diver} \cdot A + C_{in} \cdot F_{in}}{F_{out}} \cdot x_{10} \cdot E' + x_{2,i-1}^{on} \cdot E + N_{diver} \cdot A \cdot (t_{cycle} - t_{vent})$$

In order to simplify this, let

$$G = E'*x_{10}*\frac{N_{diver} \cdot A + C_{in} \cdot F_{in}}{F_{out}}$$

and

$$H = N_{diver}*A*(t_{cycle} - t_{vent}).$$

Then

$$x_{2,i}^{off} = G + E \cdot H + E \cdot x_{2,i-1}^{off}$$

$$x_{2,i}^{on} = G + H + E \cdot x_{2,i-1}^{on}$$

Both of these equations are of the form

$$f_i = a \cdot f_{i-1} + b$$

Which can be rewritten in terms of the initial value f_0

$$f_i = a^i \cdot f_0 + \frac{1-a^i}{1-a} \cdot b$$

We can determine the asymptotic value for this expression by letting the number of cycles go to infinity. Because $E < 1$, $a^i = E^i \rightarrow 0$, and

$$x_{2,\infty}^{off} = \frac{G+E \cdot H}{1-E}$$
$$x_{2,\infty}^{on} = \frac{G+H}{1-E}$$

the difference between these two is simply $H = N_{diver} * A * (t_{cycle} - t_{vent})$, which is the amount of CO_2 produced by the divers during the nonventing period.

Using this expression for $x_{2,\infty}^{on}$, and assuming we know the upper limit of CO_2 partial pressure (either 0.015 ATA, or possibly 0.02 ATA), we can determine the venting time necessary to maintain CO_2 partial pressure below this value by solving for t_{vent} in the transcendental equation

$$(1-E) \cdot P_{2,\infty}^{on} = \left((1-E) \cdot x_{10} \cdot \frac{N_{diver} \cdot A + C_{in} \cdot F_{in}}{F_{out}} + N_{diver} \cdot A \cdot (t_{cycle} - t_{vent}) \right) \cdot \frac{R_g \cdot T}{V}$$

where $P_{2,\infty}^{on}$ is the CO_2 partial pressure limit.

Appendix A contains the source code for program TABLE.FOR, which uses the above equation to solve for venting times in each of the 4 gas volumes present during DDS operations. These 4 gas volumes are the chamber, the trunk, the hanger, and the bubble. The bubble is the forward portion of the hanger that remains gas-filled when the rest of the hanger has been fully flooded. It contains the controls for venting the hanger, and is separated from the rest of the hanger by a plexiglass partition that is attached to the upper portion of the hanger. Also included in Appendix A are a sample input file for the program, and 4 sample venting tables generated by the program. Note that the program uses an IMSL FORTRAN subroutine DZBREN to find the venting times from the transcendental equation. Other equation-solving routines can be used in its place.

Appendix B contains the source code for program MODEL.FOR, which uses the entire set of equations above to model expected CO_2 concentrations during the operation of a DDS. Also included are sample input and output files, and some comparisons to data from actual DDS operations. Note that this program has been designed to be used only for the transfer trunk, since this compartment is the one for which the most detailed operational data is available.

APPENDIX A. Source code for program TABLE.FOR

```
**********

ccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
cThis program will generate DDS venting tables for the four
cdifferent areas of the DDS; the hanger, the bubble, the trunk,
cand the chamber. It is assumed that the air in each of these
ccompartments is well mixed. It is further assumed that the trunk
cis half flooded, gas space volume = 120 cubic feet, the hanger
cis unflooded during venting, gas space volume = 1400 cu ft, the
cbubble is half flooded, gas space volume = 120 cu ft, and that
cthe chamber is unflooded, gas space volume = 170 cu ft. The
ctemperature is 290 K = 63 F, and the flow rates at different
ccompartmental pressures are 38 SCFM from 0-9 fsw, 160 SCFM between
c10-29 fsw, and 240 SCFM at greater depths. The input parameters are:
c
c  ichamb     =     compartment of interest
c                1 = hanger     output in HANGER.TAB
c                2 = trunk      output in TRUNK.TAB
c                3 = chamber    output in CHAMBER.TAB
c                4 = bubble     output in BUBBLE.TAB
c     A =        oxygen consumption rate in liters/minute
c pco2max     =     CO2 limiting pressure in ATM
c   cppm      =     CO2 concentration in ppm in venting gas
c   tcycle    =     venting cycle time
ccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc

        implicit real*8 (a-h,o-z)
        real*8 k,ndart,nda
        dimension vtable(4,20),depth(4),imin(4,20),isec(4,20),flows(4)
        external dzbren,func
        common/comf/cinf,dinf,k,tcycle
c
c  set constants
c
        pi = 4.d0*datan(1.d0)
        ftm = 2.54d0*12.d0/100.d0
        ftm3 = ftm*ftm*ftm
        rgas = 8.31d0  ! units of Joule/mol*K
        temp = 290.d0     ! in K
        p0 = 1.d0         ! atm
        rgt = rgas*temp
        p0pa = 1.01325d5  ! one atm in units of pascal
c
c  get parameters for run
c
        open(unit=3, file='table.inp', type='old')
        read(3,100)ichamb,a,pco2max,cppm,tcycle
        close(3)
100     format(i12/8(f12.7/))
        co2limit=pco2max
c
c  convert  to proper units
```

```
c
        pco2max = pco2max*p0pa
        aaa=a
        tcycle = tcycle*60.d0
c  convert a from liter/min to mol/sec
        a = a/(rgas*temp)/60.d0/1000.d0*p0pa
c
c set up arrays for printing of table. depths are from current tables
c
        depth(1)=9.d0
        flows(1)=38.d0
        depth(2)=29.d0
        flows(2)=160.d0
        depth(3)=100.d0
        if(ichamb.eq.1)depth(3)=66.d0
        flows(3)=240.d0
        depth(4)=165.d0
        if((ichamb.eq.1).or.(ichamb.eq.4))depth(4)=130.d0
        flows(4)=240.d0
c
c  convert compartment volume to cubic meters, set maximum number
c  of divers for the compartment, and open the output file
c
        if(ichamb.eq.1) then
          vol = 1400.d0*ftm3
          ndmax = 20
         open(unit=4, file='hanger.tab',type='new')
        else if(ichamb.eq.2)then
          vol = 120.d0*ftm3
          ndmax = 8
          open(unit=4, file='trunk.tab',type='new')
        else if(ichamb.eq.3)then
          vol = 170.d0*ftm3
          ndmax = 8
          open(unit=4, file='chamber.tab',type='new')
        else if(ichamb.eq.4)then
          vol = 120.d0*ftm3
          ndmax = 4
          open(unit=4, file='bubble.tab',type='new')
        else
          write(6,110)
110       format('  improper compartment number -  stop program')
          stop
        end if
c
        cppm = cppm/1.d6
        errabs = 0.d0
        errrel = 1.d-6
        do 10 id=1,4
          depthdds=depth(id)
c  convert flows from cu ft/min to mol/sec
          flow=flows(id)*ftm3*p0pa/rgt/60.d0
c
c  calculate ambient pressure (in pascal), total moles of gas in
c  the compartment, and the time constant for the compartment
```

```
c
          pamb = (depthdds/33.d0 + 1.d0)*p0pa
          x10 = pamb*vol/rgt
          k = flow/x10
c
          do 20 nd=1,ndmax
            ndart = nd*a*rgt
            nda = nd*a
c
c  calculate coefficients for equation in the DDS report
c
        c0 = (1 - E) cinf + dinf*(tcycle - tvent)
c
c  where E = dexp(-k*tvent)
c
            dinf = ndart/p0pa/vol
            cinf = (pamb*(nda+cppm*flow)/flow - pco2max)/p0pa
            tlow = 0.d0
            thigh = 1.d4
             maxfn = 10000
c
c  dzbren is an IMSL FORTRAN root finding routine
c
             call dzbren(func,errabs,errrel,tlow,thigh,maxfn)
             vtable(id,nd)=thigh
             imin(id,nd)=thigh/60.d0
             isec(id,nd)=thigh-60.d0*imin(id,nd)+1
20        continue
10      continue
        volf=vol/ftm3
        jmax=ndmax
        if(ichamb.eq.1)then
          write(4,181)volf,aaa,co2limit
181       format(/' HANGER, unflooded, volume = ',f6.1,' cu. ft.'/
     1 ' O2 consumption = ',f5.2,' liter/min'/
     2 ' CO2 limit = ',f6.3,' atm'/)
        else if(ichamb.eq.4)then
          write(4,184)volf,aaa,co2limit
184       format(/' BUBBLE, water depth = 4.5 ft, volume = ',f6.1,
     1  ' cu. ft.'/
     2 ' O2 consumption = ',f5.2,' liter/min'/
     3 ' CO2 limit = ',f6.3,' atm'/)
        else if(ichamb.eq.2)then
          write(4,182)volf,aaa,co2limit
182       format(/' TRUNK, water depth = 4.5 ft, volume = ',f6.1,
     1  ' cu. ft.'/
     2 ' O2 consumption = ',f5.2,' liter/min'/
     3 ' CO2 limit = ',f6.3,' atm'/)
        else if(ichamb.eq.3)then
          write(4,183)volf,aaa,co2limit
183       format(/' CHAMBER, unflooded, volume = ',f6.1,' cu. ft.'/
     1 ' O2 consumption = ',f5.2,' liter/min'/
     2 ' CO2 limit = ',f6.3,' atm'/)
        end if
        if(ichamb.eq.1)write(4,191)
```

```
      if(ichamb.eq.2)write(4,190)
      if(ichamb.eq.3)write(4,190)
      if(ichamb.eq.4)write(4,194)
190   format(/' # Divers',16x,'DEPTH(FSW)'/
    1 '         0 - 9     10 - 29    30 - 100   101 - 165'/,
    2 '       38 SCFM    160 SCFM    240 SCFM    240 SCFM'/)
191   format(/' # Divers',16x,'DEPTH(FSW)'/
    1 '         0 - 9     10 - 29    30 - 66    67 - 130'/,
    2 '       38 SCFM    160 SCFM    240 SCFM    240 SCFM'/)
194   format(/' # Divers',16x,'DEPTH(FSW)'/
    1 '         0 - 9     10 - 29    30 - 100   101 - 130'/,
    2 '       38 SCFM    160 SCFM    240 SCFM    240 SCFM'/)
      write(4,200)(j,(imin(l,j),isec(l,j),l=1,4),j=1,jmax)
200   format(20(2x,i3,5x,4(i3,2h' ,i2,1h",5x)/))
      stop
      end
c
      real*8 function func(x)
      implicit real*8 (a-h,o-z)
      common/comf/c,d,xk,xc
      func = (1.d0 - dexp(-xk*x))*c + d*(xc - x)
      return
      end
c
```

****

End of souce code for program TABLE.FOR

Sample input file TABLE.INP for program TABLE.FOR

```
**********
     3
 1.25d0
 0.015d0
 1250.d0
  5.d0
```

read(3,100)ichamb,a,pco2max,cppm,tcycle

****

End of sample input file TABLE.INP

Table 1. Sample Output File HANGER.TAB From Program TABLE.FOR				
	Depth (FSW)			
Diver No.	*0 – 9* *38 SCFM*	*10 – 29* *160 SCFM*	*30 – 66* *240 SCFM*	*67 – 130* *240 SCFM*
1	0'34"	0'13"	0'15"	0'32"
2	1' 7"	0'25"	0'30"	1' 3"
3	1'41"	0'38"	0'45"	1'35"
4	2'15"	0'51"	1' 1"	2' 7"
5	2'48"	1' 4"	1'16"	2'38"
6	3'21"	1'16"	1'31"	3'10"
7	3'54"	1'29"	1'47"	3'40"
8	4'27"	1'42"	2' 2"	4'11"
9	4'58"	1'55"	2'18"	4'40"
10	5'29"	2' 8"	2'33"	5' 9"
11	5'60"	2'21"	2'48"	5'36"
12	6'29"	2'34"	3' 3"	6' 3"
13	6'58"	2'46"	3'18"	6'28"
14	7'25"	2'59"	3'33"	6'53"
15	7'52"	3'12"	3'47"	7'16"
16	8'18"	3'24"	4' 1"	7'39"
17	8'43"	3'36"	4'15"	7'60"
18	9' 7"	3'48"	4'29"	8'20"
19	9'30"	3'60"	4'42"	8'40"
20	9'52"	4'12"	4'56"	8'58"

Hanger, water depth = 0.0 volume = 1400.0
O_2 consumption = 1.25 liter/min
SCFM = standard cubic feet per minute
Tcycle = 5 min

Table 2. Sample output file TRUNK.TAB from program TABLE.FOR				
	Depth (FSW)			
Diver No.	*0 - 9* *38 SCFM*	*10 – 29* *160 SCFM*	*30 – 100* *240 SCFM*	*101 – 165* *240 SCFM*
1	0'36"	0'14"	0'25"	0'50"
2	1'15"	0'29"	0'54"	1'47"
3	1'55"	0'47"	1'26"	2'44"
4	2'36"	1' 7"	2' 1"	3'33"
5	3'13"	1'28"	2'34"	4'12"
6	3'46"	1'51"	3' 4"	4'42"
7	4'14"	2'14"	3'29"	5' 5"
8	4'38"	2'35"	3'51"	5'23"

Trunk, water depth = 0.0 volume = 120.0; O_2 consumption = 1.25 liter/min *Tcycle = 5 min*

Table 3. Sample output file CHAMBER.TAB from program TABLE.FOR				
	Depth (FSW)			
Diver No.	*0 – 9* *38 SCFM*	*10 – 29* *160 SCFM*	*30 – 100* *240 SCFM*	*101 - 165* *240 SCFM*
1	0'35"	0'13"	0'24"	0'48"
2	1'12"	0'28"	0'51"	1'41"
3	1'50"	0'43"	1'20"	2'34"
4	2'28"	0'60"	1'50"	3'23"
5	3' 5"	1'18"	2'20"	4' 4"
6	3'38"	1'37"	2'48"	4'38"
7	4' 8"	1'56"	3'14"	5' 6"
8	4'35"	2'15"	3'37"	5'28"

CHAMBER, unflooded, volume = 170.0 cu. ft.; O_2 consumption = 1.25 liter/min; CO_2 limit = 0.015 atm Tcycle = 5 min

Table 4. Sample output file BUBBLE.TAB from program TABLE.FOR				
	Depth (FSW)			
Diver No.	*0 – 9* *38 SCFM*	*10 – 29* *160 SCFM*	*30 – 100* *240 SCFM*	*101 – 130* *240 SCFM*
1	0'36"	0'14"	0'25"	0'35"
2	1'15"	0'29"	0'54"	1'15"
3	1'55"	0'47"	1'26"	1'58"
4	2'36"	1' 7"	2' 1"	2'41"

BUBBLE, water depth = 4.5 ft, volume = 120.0 cu. ft.; O_2 consumption = 1.25 liter/min; CO_2 limit = 0.015 atm Tcycle = 5 min

APPENDIX B. Source code for program MODEL.FOR

```
**********


ccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
c
c  This program will model the CO2 concentration in a dry deck
c  shelter (DDS) under the following assumptions:
c
c       1       parameters include the pressure in the compartment,
c               the depth of water in the compartment, and the
c               number of divers in the compartment.
c       2       fixed parameters include the O2 consumption rate,
c               the CO2 level in the venting gas, the initial CO2
c               level at time t = 0.
c       3       the maximum flow rate is depth dependent
c                       38 SCFM from 0-9 fsw
c                       160 SCFM from 10-29 fsw
c                       240 SCFM   > 30 fsw
c       4       changes in compartment volume or pressure,
c               or venting of the compartment do not occur
c               simultaneously.
c
cccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc

      implicit real*8 (a-h,o-z)
      real*8 nda,nco2,ngas
cccc
      dimension tmin(500),fsw(500),dwater(500),ndiver(500),ivent(500)
      dimension vol(500),pamb(500),dpambdt(500),tsec(500),dvdt(500)
      dimension fin(500),fout(500),nco2(500),ngas(500),pc02(500)

      common/dds0/rtrunk,pi
      common/comvent/pamb010,pamb1030,vent010,vent1030,vent30

      xnil=1.d-10
      zed=0.d0

      open (unit=1,status='old',name='model.inp')
      open (unit=2,status='new',name='model.out')

      rg=82.0562d0        ! gas constant, (cc-atm)/(gmol-degK)
      rgas=8.31            ! gas constant, J/(gmol-degK)
      rgl=rg/1000.d0
      rgm=rg/1.d6
      rgf=2.89787d-3      ! gas constant, (ft3-atm)/(gmol-degK)
      temp=290.d0
      temp0=273.16d0
      rgast=rgas*temp
      ftm = 2.54d0*12.d0/100.d0
      pi=4.d0*datan(1.d0)
c
c  estimate the radius of the trunk by using a volume of 240 cu ft
```

```
c
        rtrunk = (240.d0*3.d0/4.d0/pi)**(1.d0/3.d0)*ftm

        p0pa = 1.01325d5    ! 1atm in pascals
        p0= 1.d0            ! 1 atm pressure for reference
        fsw0=33.d0          ! 1atm in fsw
        t=0.d0              ! initializing
        pamb010=10.d0/fsw0*p0pa
        pamb1030=30.d0*p0pa
        vent010=38.d0/60.d0/rgf/temp0
        vent1030=160.d0/60.d0/rgf/temp0
        vent30=240.d0/60.d0/rgf/temp0
c
c  read in parameters
c
        read (1,*)  nnode    ! number of nodes in profile
        read (1,*)  a        ! oxygen consumption rate (liter/min)
        read (1,*)  cin      ! concentration co2 in ppm in venting gas
        read (1,*)  cout     ! initial conc. co2 in ppm in chamber
        a = a/(rgl*temp0)/60.d0  ! convert to mol/sec
        cin=cin/1.d6
        cout=cout/1.d6
c       fvent=fvent/60.d0/rgf/temp0  ! change from SCFM to mol/sec
c
c  will assume that venting lasts for the entire period from the current
c  node to the next node, assuming that venting is off by the next node.
c  this requires that when ivent(k) = 1, tmin(k+1)=tmin(k)+tvent
c
        do 10 i=1,nnode
          read(1,11)tmin(i),fsw(i),dwater(i),ndiver(i),ivent(i)
          pamb(i)=(fsw(i)+fsw0)/fsw0*p0pa
          dwater(i)=dwater(i)*ftm
          vol(i)=vtrunk(dwater(i))
          tsec(i)=tmin(i)*60.d0
10      continue
11      format(3f10.2,2i10,f10.2)
        do 12 i=1,nnode-1
          fin(i)=zed
          fout(i)=zed
          delt=tsec(i+1)-tsec(i)
c
c  assume that chamber volume and pressure do not change simultaneously
c  nor do such changes occur during venting
c
c  calculate fin and fout for changes in ambient pressure
c
          dpambdt(i)=(pamb(i+1)-pamb(i))/delt
          if(dpambdt(i).gt.1.d-8)then
            fin(i)=vol(i)*dpambdt(i)/rgast
          else if(dpambdt(i).lt.(-1.d-8))then
            fout(i)= -vol(i)*dpambdt(i)/rgast
          end if
c
c  calculate fin and fout for changes in volume
c
```

```
            dvdt(i)=(vol(i+1)-vol(i))/delt
            if(dvdt(i).gt.1.d-8)then
              fin(i)=pamb(i)*dvdt(i)/rgast
            else if(dvdt(i).lt.(-1.d-8))then
              fout(i)= -pamb(i)*dvdt(i)/rgast
            end if
c
c  calculate fin and fout for venting
c
            if(ivent(i).eq.1)then
              fin(i)=fventfcn(pamb(i))
              fout(i)=fin(i)
            end if
12        continue
          tout=0.d0
          delt=60.d0
c
c  calculate the co2 partial pressures at each of the nodes
c  in the profile
c
          ngas(1)=pamb(1)*vol(1)/rgast
          nco2(1)=cout*ngas(1)
          do 200 leg=1,nnode-1
            nda=ndiver(leg)*a
            fint=fin(leg)
            foutt=fout(leg)
            fnet=fint-foutt
            cnet=foutt
            bnet=cin*fint+nda
            delt=tsec(leg+1)-tsec(leg)
            x10=ngas(leg)
            x20=nco2(leg)
c
c  cases (a) & (b):  fout=c=0,
c
            if(dabs(cnet).lt.1.d-8)then
              ngas(leg+1)=f1case(x10,fnet,delt)
              nco2(leg+1)=f2casea(x20,bnet,delt)
            end if
c
c  case (c):  fout=c>0, f=0 -> fin=fout
c
            if((dabs(cnet).ge.1.d-8).and.(dabs(fnet).lt.1.d-8))then
              ngas(leg+1)=x10
              nco2(leg+1)=f2casec(x20,x10,bnet,cnet,delt)
            end if
c
c  case (d):  fout=c>0, fin > 0, f <> 0
c
            if((dabs(cnet).ge.1.d-8).and.(dabs(fnet).ge.1.d-8))then
              if(dabs(fint).lt.1.d-8)then
                    ngas(leg+1)=x10-cnet*delt
                    nco2(leg+1)=f2cased1(x20,x10,bnet,cnet,delt)
              else if(dabs(fint).ge.1.d-1)then
                    ngas(leg+1)=x10+fnet*delt
```

```
                nco2(leg+1)=f2cased3(x20,x10,bnet,cnet,fnet,delt)
           end if
         end if
200      continue
         do 201 i=1,nnode
           pco2=nco2(i)*rgast/vol(i)
           tout=tsec(i)/60.d0
           pgas=ngas(i)*rgast/vol(i)
c
c  convert to atm and ppm for output
c
           pco2=pco2/p0pa
           pgas=pgas/p0pa
           cppm=nco2(i)/ngas(i)*1.d6
c
c  want cppm as surface equivalent
c
           cppmsev=nco2(i)/ngas(i)*1.d6*pgas
           write(2,111)tout,pco2,pgas,cppmsev
           write(6,111)tout,pco2,pgas,cppmsev
201      continue
111      format(3f11.5,3x,f11.2,i10)

30       stop
         end

CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
CCCCCCCCCC

         real*8 function vtrunk(d)
         implicit real*8 (a-h,o-z)
         common/dds0/rtrunk,pi
         vtrunk = pi/3.d0*(2.d0*rtrunk-d)**2.d0*(rtrunk+d)
         return
         end

CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
CCCCCCCCCC

         real*8 function f1case(x10,flow,delt)
         implicit real*8 (a-h,o-z)
         f1case=x10+flow*delt
         return
         end

CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
CCCCCCCCCC

         real*8 function f2casea(x20,slope,delt)
         implicit real*8 (a-h,o-z)
         f2casea=x20+slope*delt
         return
         end
```

```
cccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
cccccccccc

        real*8 function f2casec(x20,x10,b,c,delt)
        implicit real*8 (a-h,o-z)
        expterm=dexp(-c/x10*delt)
        f2casec=b/c*x10*(1.d0-expterm)+x20*expterm
        return
        end

cccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
cccccccccc

        real*8 function f2cased1(x20,x10,b,c,delt)
        implicit real*8 (a-h,o-z)
        x1=f1case(x10,-c,delt)
        f2cased1=(x20/x10+b/c*dlog(x10/x1))*x1
        return
        end

cccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
cccccccccc

        real*8 function f2cased3(x20,x10,b,c,f,delt)
        implicit real*8 (a-h,o-z)
        x1=x10+f*delt
        f2cased3=b/(f+c)*x1-(b*x10/(f+c)-x20)*((x10/x1)**(c/f))
        return
        end

cccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
cccccccccc

        real*8 function fventfcn(pamb)
        implicit real*8 (a-h,o-z)
        common/comvent/pamb010,pamb1030,vent010,vent1030,vent30

        if(pamb.lt.pamb010)then
          fventfcn=vent010
        else if(pamb.lt.pamb1030)then
          fventfcn=vent1030
        else
          fventfcn=vent30
        end if
        return
        end

c
**********
```

End of source code for program MODEL.FOR

Sample input file MODEL.INP for program MODEL.FOR

```
**********
    31,  ! number of nodes in profile
  0.85d0,           ! oxygen consumption  rate in liter/min/diver
   10.d0,! co2 conc. in ppm in venting gas
 6300.d0,           ! initial co2 conc. in ppm in chamber
   0.d0,   0.d0,   0.d0,   4,  0,
   4.d0,   0.d0,   0.d0,   4,  1,
 5.18d0,   0.d0,   0.d0,   4,  0,
   8.d0,   0.d0,   4.d0,   4,  0,
   9.d0,   0.d0,   4.d0,   4,  1,
10.18d0,   0.d0,   4.d0,   4,  0,
  12.d0,   0.d0,   4.d0,   4,  0,
  14.d0,  20.d0,   4.d0,   4,  0,
  15.d0,  30.d0,   4.d0,   4,  0,
  16.d0,  30.d0,   4.d0,   4,  0,
  21.d0,  30.d0,   4.d0,   4,  0,
  25.d0,  30.d0,   0.d0,   4,  1,
  26.d0,  30.d0,   0.d0,   4,  0,
  28.d0,  30.d0,   0.d0,   4,  0,
  31.d0,  30.d0,   0.d0,   4,  1,
  32.d0,  30.d0,   0.d0,   4,  0,
  33.d0,  30.d0,   0.d0,   4,  0,
  36.d0,  30.d0,   0.d0,   4,  1,
  37.d0,  30.d0,   0.d0,   4,  0,
  42.d0,   0.d0,   0.d0,   4,  1,
43.18d0,   0.d0,   0.d0,   4,  0,
  47.d0,   0.d0,   0.d0,   4,  0,
48.18d0,   0.d0,   0.d0,   4,  0,
  50.d0,   0.d0,   4.d0,   4,  0,
  53.d0,  30.d0,   4.d0,   4,  0,
  54.d0,  30.d0,   4.d0,   4,  0,
  55.d0,  30.d0,   4.d0,   4,  0,
  57.d0,  30.d0,   4.d0,   4,  0,
  58.d0,  30.d0,   4.d0,   4,  1,
  59.d0,  30.d0,   4.d0,   4,  0,
  60.d0,  30.d0,   4.d0,   4,  0,

        read (1,*)  nnode    ! number of nodes in profile
        read (1,*)  a        ! oxygen consumption rate
        read (1,*)  cin      ! concentration co2 in ppm in venting gas
          read(1,11)tmin(i),fsw(i),dwater(i),ndiver(i),ivent(i)

**********
```

End of sample input file MODEL.INP

The next several pages include an actual input data file (TM071797.INP) that was used for the program MODEL.FOR, the data sheets from which the input file was derived, the ouput file (TO071797.OUT), and a plot of the calculated CO_2 levels along with the values measured using the CO_2 analyzer. This particular data set was chosen due to its richness. The person recording the data did an excellent job, especially in recording the duration of the venting periods. Note that there are several uncertainties in the data: (1) The depth of water in the trunk is an estimate, (2) the total volume of the unflooded trunk is estimated and does not take into account the space taken up by personnel or equipment that is being transfered, (all of which lead to uncertainty in the gas volume), and (3) the concentration of CO_2 in the venting gas is unknown. For this particular input file, CO_2 was assumed to be present at a very low level of 10 ppm.

Data file TO071797.INP

```
**********
    66,  ! number of nodes in profile
  0.85d0,            ! oxygen consumption  rate in liter/min/diver
   10.d0, ! co2 conc. in ppm in venting gas
  4650.d0,           ! initial co2 conc. in ppm in chamber
0.,0.,0.,2,1,4650,0.77
0.77,0.,0.,2,0,,
6.,0.,0.,2,1,4340,0.77
6.77,0.,0.,2,0,,
14.,0.,3.,2,1,3960,0.33
14.33,0.,3.,2,0,,
21.,0.,3.5,2,1,6040,0.68
21.68,0.,3.5,2,0,,
29.,0.,5.5,2,1,9220,0.77
29.77,0.,5.5,2,0,,
36.,0.,5.5,2,1,7820,0.65
36.65,0.,5.5,2,0,,
43.,0.,5.5,2,1,7060,0.82
43.82,0.,5.5,2,0,,
49.,0.,5.5,2,1,6640,0.78
49.78,0.,5.5,2,0,,
57.,0.,5.5,2,1,7280,0.75
57.75,0.,5.5,2,0,,
64.,5.,5.5,2,1,7660,0.8
64.8,5.,5.5,2,0,,
72.,5.,5.5,2,1,7680,0.52
72.52,5.,5.5,2,0,,
79.,5.,5.5,2,1,8200,0.82
79.82,5.,5.5,2,0,,
85.,5.,5.5,2,1,6140,0.52
85.52,5.,5.5,2,0,,
92.,5.,5.5,2,1,6420,0.55
92.55,5.,5.5,2,0,,
99.,5.,5.5,2,1,7000,0.97
99.97,5.,5.5,2,0,,
105.,5.,5.5,2,1,4580,0.63
105.63,5.,5.5,2,0,,
111.,5.,5.5,2,1,5240,0.77
111.77,5.,5.5,2,0,,
117.,5.,5.5,2,1,4800,0.63
117.63,5.,5.5,2,0,,
123.,5.,5.5,2,1,4840,0.73
123.73,5.,5.5,2,0,,
129.,5.,5.5,2,1,6880,0.7
129.7,5.,5.5,2,0,,
136.,5.,5.5,2,1,5900,0.92
136.92,5.,5.5,2,0,,
142.,5.,5.5,2,1,3800,0.45
142.45,5.,5.5,2,0,,
150.,5.,5.5,2,1,5700,1.05
151.05,5.,5.5,2,0,,
156.,5.,5.5,2,1,5200,0.67
156.67,5.,5.5,2,0,,
```

```
164.,5.,5.5,2,1,7000,0.43
164.43,5.,5.5,2,0,,
170.,6.,5.5,2,1,7400,0.65
170.65,6.,5.5,2,0,,
175.,6.,5.5,2,1,4720,0.63
175.63,6.,5.5,2,0,,
182.,6.,5.5,2,1,5840,0.58
182.58,6.,5.5,2,0,,
189.,6.,5.5,2,1,5600,0.47
189.47,6.,5.5,2,0,,
194.,6.,5.5,2,1,6000,0.58
194.58,6.,5.5,2,0,,
204.,6.,5.5,2,1,5840,0.58
204.58,6.,5.5,2,0,,
215.,30.,5.5,2,0,11380,
216.,40.,5.5,2,0,13240,
219.,50.,5.5,2,0,15160
221.,60.,5.5,2,0,17780
**********
End of file TO071797.INP
```

Output data file TM071797.DAT

****

.00000	.00465	1.00000	4650.00
.77000	.00286	1.00000	2858.91
6.00000	.00425	1.00000	4247.10
6.77000	.00263	1.00000	2625.22
14.00000	.00496	1.00000	4964.50
14.33000	.00361	1.00000	3606.55
21.00000	.00649	1.00000	6485.02
21.68000	.00300	1.00000	3004.06
29.00000	.00852	1.00000	8516.59
29.77000	.00091	1.00000	914.85
36.00000	.00921	1.00000	9207.00
36.65000	.00126	1.00000	1264.69
43.00000	.00972	1.00000	9716.56
43.82000	.00089	1.00000	894.23
49.00000	.00779	1.00000	7788.83
49.78000	.00085	1.00000	850.64
57.00000	.01046	1.00000	10460.49
57.75000	.00109	1.00000	1089.71
64.00000	.00941	1.15152	9409.99
64.80000	.00121	1.15152	1206.31
72.00000	.01079	1.15152	10789.54
72.52000	.00253	1.15152	2528.59
79.00000	.01115	1.15152	11153.49
79.82000	.00130	1.15152	1300.19
85.00000	.00819	1.15152	8194.79
85.52000	.00201	1.15152	2005.67
92.00000	.01063	1.15152	10630.58
92.55000	.00232	1.15152	2315.33
99.00000	.01090	1.15152	10900.31
99.97000	.00097	1.15152	970.48
105.00000	.00767	1.15152	7665.42
105.63000	.00148	1.15152	1480.70
111.00000	.00863	1.15152	8628.19
111.77000	.00121	1.15152	1207.22
117.00000	.00817	1.15152	8168.36
117.63000	.00155	1.15152	1552.92
123.00000	.00870	1.15152	8700.41
123.73000	.00131	1.15152	1314.97
129.00000	.00833	1.15152	8329.36
129.70000	.00136	1.15152	1356.39
136.00000	.00974	1.15152	9741.71
136.92000	.00099	1.15152	990.11
142.00000	.00775	1.15152	7751.61
142.45000	.00227	1.15152	2270.77
150.00000	.01232	1.15152	12319.85
151.05000	.00091	1.15152	911.30
156.00000	.00750	1.15152	7499.77
156.67000	.00134	1.15152	1339.44
164.00000	.01110	1.15152	11095.69
164.43000	.00328	1.15152	3276.12
170.00000	.01069	1.18182	10690.11
170.65000	.00191	1.18182	1910.04

175.00000	.00770	1.18182	7699.91
175.63000	.00155	1.18182	1548.73
182.00000	.01003	1.18182	10027.22
182.58000	.00213	1.18182	2133.92
189.00000	.01068	1.18182	10678.97
189.47000	.00295	1.18182	2949.61
194.00000	.00898	1.18182	8979.05
194.58000	.00195	1.18182	1950.11
204.00000	.01449	1.18182	14488.16
204.58000	.00292	1.18182	2916.25
215.00000	.01679	1.90909	16792.58
216.00000	.01813	2.21212	18126.61
219.00000	.02212	2.51515	22122.65
221.00000	.02479	2.81818	24787.69

****

End of output file TM071797.DAT

Following are the two raw data sheets, as well as a graph of the measured and calculated CO_2 levels. Bitmap copies of the data sheets are contained in files DDSDATA1.PDF. The graph is contained in a Hewlett-Packard Graphics Language file TM071797.PDF.

1

CO2 Analyzer Test Data Sheet
(Use New Sheet Each Day)

Analyzer Serial#: 135
Cal Gas % CO2: 1.5 %
Cal Gas Cylinder#: BC 105604
Cal Gas Reading: 16,400 %
Time checked: 1200
Re-calibrated: Yes ✓ No

Date: 17 Jul 97
Submarine: USS Bates

SPHERE: TRUNK

Time	Presure (fsw)	Depth of Water in Sphere (fsw)	Time of EVERY Vent	# of Divers	Diver Activity	CO2 (ppm) ANALYZER READING	Cylinder # (If take sample)
2122	0'	0'	0.77 :46	2	REST	4650	—
2128	Ø	Ø	0.77 :46			4346	—
2136	Ø	3'	0.33 :20			3960	* 63
2143	Ø	3.5	0.68 :41			6046	—
2151	Ø	5.5'	0.77 :46			9220	—
2158	Ø	5.5'	0.65 :39			7820	—
2205	Ø	5.5'	0.82 :49			7060	—
2211	Ø	5.5'	0.78 :47			6640	—
2219	Ø	5.5'	0.75 :45			7280	44
2226	5'	5.5'	0.80 :48			7660	—
2234	5'	5.5'	0.52 :31			7680	—
2241	5'	5.5'	0.82 :49			8200	—
2247	5'	5.5'	0.52 :31			6140	—
2254	5'	5.5'	0.55 :33			6420	—
2301	5'	5.5'	0.97 :58			7000	—
2307	5'	5.5'	0.63 :38			4580	—
2313	5'	5.5'	0.77 :46			5240	—
2319	5'	5.5'	0.63 :38			4800	—

Notes: * #63 APPEARED TO HAVE LESS VACUUM VOLUME THAN THE OTHER CYLINDERS AS PER TESTOR'S PERSONAL OPINION

CO2 Analyzer Test Data Sheet

(Use New Sheet Each Day)

Analyzer Serial#: 135
Cal Gas % CO2: 1.5 %
Cal Gas Cylinder#: BC105604
Cal Gas Reading: 16,400 %
Time checked: 1200
Re-calibrated: Yes ✓ No____

Date: 17 JUL 97
Submarine: USS BATES

SPHERE: TRUNK

Time	Presure (fsw)	Depth of Water in Sphere (fsw)	Time of EVERY Vent	# of Divers	Diver Activity	CO2 (ppm) ANALYZER READING	Cylinder # (If take sample)
2325	5'	5.5'	0.73 :44	2	REST	4840	—
2331	5'	5.5'	0.70 :42			6880	—
2338	5'	5.5'	0.92 :55			5900	—
2344	5'	5.5'	0.75 :45			3800	—
2352	5'	5.5'	1.05 1:03			5700	—
2358	5'	5.5'	0.67 :40			5200	—
0006	5'	5.5'	0.43 :26			7000	—
0012	6'	5.5'	0.65 :34			7400	—
0017	6'	5.5'	0.63 :38			4720	—
0024	6'	5.5'	0.58 :35			5840	—
0031	6'	5.5'	0.47 :28			5600	—
0036	6'	5.5'	0.56 :35			6000	—
0046	6'	5.5'	0.58 :35			5840	—
0107	30'	5.5'	PRESS			11380	15
0108	40'	5.5'	PRESS			13240	24
0111	50'	5.5'	PRESS			15160	14
0113	60'	5.5'	PRESS			17780	34

Notes:

FIGURE 1



CO2 Concentration (ppm)
25000
20000
15000
10000
5000
0
0
50
100
150
200
250
Time (min)